
Summary

Microsoft Business Solutions products and Microsoft Office are increasingly integrated. Using web services, it is possible to invoke Navision business processes while working in the Microsoft Office product. In the following example:

· You can create a letter template in Microsoft Word to remind a customer that he reached his credit limit in Navision
· You can retrieve from Microsoft Word (without opening Navision) the customer’s details to fill the letter template automatically (Address, Contact Names…)
· You can block the customer from Microsoft Word (without opening Navision), so no additional sales orders can be created in Navision without approval
You need to install the following components:

· Microsoft Windows XP SP2 (Professional Edition) – Version 5.1.2600
· Microsoft Office Word 2003 SP2 (Professional Edition) – Version 11.6568.6568
· Microsoft Visual Studio 2005 (Professional Edition) – Version 8.0.50727.42
· Microsoft.NET Framework – Version 2.0.50727
· Microsoft Office 2003 Web Services Toolkit (http://msdn.microsoft.com/library/default.asp?url=/library/en-us/odc_wst2003_ta/html/odc_WSTKNew.asp)
· Microsoft Message Queuing - Version: 5.01.1108 (Control Panel / Add or Remove Programs / Add/Remove Windows Components)
· Microsoft Navision 4.00 SP1 – Version 4.0.1.21666
· Microsoft Navision Application Server 4.00 SP1 – Version 4.0.1.21666 (refer to the Navision Application Server documentation included in the Navision product CD - Doc\w1w1atas.pdf)
The demo contains 3 steps:

· Configuring a Navision Application Server to access Navision

· Creating a web service in Visual Studio 2005

· Calling the web service from Microsoft Word (Customer Credit Limit.doc embedded)
Configuring a Navision Application Server to access Navision

Create and compile XMLport 80000 Customer by copy/ pasting the following C/AL code lines in a text file and import into Navision

OBJECT XMLport 80000 Customer

{

  OBJECT-PROPERTIES

  {

    Date=;

    Time=;

    Version List=NavisionWebService;

  }

  PROPERTIES

  {

  }

  ELEMENTS

  {

    { [{BF00F684-48BE-4FA3-9A60-D83313729921}];  ;Customer            ;Element ;Table   ;

                                                  VariableName=Customer;

                                                  SourceTable=Table18;

                                                  MaxOccurs=Once }

    { [{36E410F8-09CA-432C-A2D7-9EEB24513092}];1 ;No                  ;Element ;Field   ;

                                                  DataType=Code;

                                                  SourceField=Customer::No. }

    { [{449E6B20-B2AE-4585-BF5D-547100835006}];1 ;Name                ;Element ;Field   ;

                                                  DataType=Text;

                                                  SourceField=Customer::Name }

    { [{0C886C80-C7A9-4E81-AE78-81DE61832033}];1 ;Address             ;Element ;Field   ;

                                                  DataType=Text;

                                                  SourceField=Customer::Address }

    { [{1B4BBDE2-33BA-45B4-93E4-C8E008C45874}];1 ;Address2            ;Element ;Field   ;

                                                  DataType=Text;

                                                  SourceField=Customer::Address 2 }

    { [{6393F10D-A0FF-4E7B-AE81-E6A984B1ED85}];1 ;City                ;Element ;Field   ;

                                                  DataType=Text;

                                                  SourceField=Customer::City }

    { [{D2B42495-00F6-4906-A4B1-2A5E73F7254C}];1 ;Contact             ;Element ;Field   ;

                                                  DataType=Text;

                                                  SourceField=Customer::Contact }

    { [{BAFCD676-8601-4144-94EA-56BA5B202A5D}];1 ;PhoneNo             ;Element ;Field   ;

                                                  DataType=Text;

                                                  SourceField=Customer::Phone No. }

    { [{01C9BE54-E4E6-4949-B964-9A0430D4A2C6}];1 ;CreditLimitLCY      ;Element ;Field   ;

                                                  DataType=Decimal;

                                                  SourceField=Customer::Credit Limit (LCY) }

    { [{81FEEBC7-59F7-4EC1-84A9-D74B26050443}];1 ;CountryCode         ;Element ;Field   ;

                                                  DataType=Code;

                                                  SourceField=Customer::Country Code }

    { [{2041A5A2-584E-47D4-9DB6-495004D419F9}];1 ;BalanceLCY          ;Element ;Field   ;

                                                  DataType=Decimal;

                                                  AutoCalcField=Yes;

                                                  SourceField=Customer::Balance (LCY);

                                                  Export::OnBeforePassField=BEGIN

                                                                              Customer.CALCFIELDS("Balance (LCY)");

                                                                            END;

                                                                             }

    { [{F2D72680-C141-4FF5-9874-ADA61EFED87A}];1 ;FaxNo               ;Element ;Field   ;

                                                  DataType=Text;

                                                  SourceField=Customer::Fax No. }

    { [{93F35292-DC24-4ABF-BFB4-1DBD85A61FF4}];1 ;PostCode            ;Element ;Field   ;

                                                  DataType=Code;

                                                  SourceField=Customer::Post Code }

    { [{D38CD468-1A4E-40B9-ABBB-92DCDBED7F34}];1 ;County              ;Element ;Field   ;

                                                  DataType=Text;

                                                  SourceField=Customer::County }

  }

  EVENTS

  {

  }

  CODE

  {

    BEGIN

    END.

  }

}
Create and compile Codeunit 80000 RunNavisionProcess by copy/ pasting the following C/AL code lines in a text file and import into Navision
OBJECT Codeunit 80000 RunNavisionProcess

{

  OBJECT-PROPERTIES

  {

    Date=;

    Time=;

    Version List=NavisionWebService;

  }

  PROPERTIES

  {

    SingleInstance=Yes;

    OnRun=BEGIN

            IF ISCLEAR(ComCom) THEN

              IF NOT CREATE(ComCom) THEN

                ERROR(Text001);

            IF ISCLEAR(MSMQ_BA) THEN

              IF NOT CREATE(MSMQ_BA) THEN

                ERROR(Text001);

            ComCom.AddBusAdapter(MSMQ_BA,1);

            MSMQ_BA.OpenReceiveQueue('.\private$\' + COMPANYNAME + ' 1',0,0);

            MSMQ_BA.OpenReplyQueue('.\private$\' + COMPANYNAME + ' 2',0,0);

          END;

  }

  CODE

  {

    VAR

      ComCom@1450001 : Automation "{F9A57667-8AC5-45C5-9416-99D3955BCAC0} 1.0:{01018FA5-E4B4-413C-A47C-AD34B0CC2647}:'Navision Communication Component version 2'.CommunicationComponent" WITHEVENTS;

      MSMQ_BA@1450000 : Automation "{B8BD635A-E191-47EF-84A0-02921E2A44A6} 1.0:{CD49794B-0E84-4A2E-9522-C518C825D390}:'Navision MS-Message Queue Bus Adapter'.MSMQBusAdapter";

      Text001@1450003 : TextConst 'ENU=Cannot create communication components.';

      Text002@1450002 : TextConst 'ENU=Document No. cannot be found';

      Text003@1450004 : TextConst 'ENU=Document Type is missing in ParameterList';

      Text004@1450005 : TextConst 'ENU=Company does not exist';

      Text005@1450006 : TextConst 'ENU=Item No. cannot be found';

      Text006@1450007 : TextConst 'ENU=Completed';

      Text007@1450008 : TextConst 'ENU=Customer No. cannot be found';

      Text008@1450009 : TextConst 'ENU=Customer has been blocked in Navision';

    LOCAL PROCEDURE ReturnOriginalMessage@1240002(OutMsg@1450002 : Automation "{F9A57667-8AC5-45C5-9416-99D3955BCAC0} 1.0:{6CB9762C-E61C-4F96-BA34-8B20D3A5B46E}:'Navision Communication Component version 2'.OutMessage";XMLDoc@1450000 : Automation "{F5078F18-C551-11D3-89B9-0000F81FE221} 3.0:{F6D90F11-9C73-11D3-B32E-00C04F990BB4}:'Microsoft XML, v3.0'.DOMDocument");

    VAR

      OutStreamQueue@1450001 : OutStream;

    BEGIN

      OutStreamQueue := OutMsg.GetStream();

      XMLDoc.save(OutStreamQueue);

      OutMsg.Send(0);

      CLEAR(OutMsg);

    END;

    LOCAL PROCEDURE SendItem@1450000(ItemNo@1450000 : Code[20];OutMsg@1450006 : Automation "{F9A57667-8AC5-45C5-9416-99D3955BCAC0} 1.0:{6CB9762C-E61C-4F96-BA34-8B20D3A5B46E}:'Navision Communication Component version 2'.OutMessage");

    VAR

      Item@1450003 : Record 27;

      ItemXMLPort@1450002 : XMLport 8002;

      OutStreamQueue@1450004 : OutStream;

      XMLDoc@1450001 : Automation "{F5078F18-C551-11D3-89B9-0000F81FE221} 3.0:{F6D90F11-9C73-11D3-B32E-00C04F990BB4}:'Microsoft XML, v3.0'.DOMDocument";

    BEGIN

      OutStreamQueue := OutMsg.GetStream();

      Item.SETRANGE("No.",ItemNo);

      ItemXMLPort.SETTABLEVIEW(Item);

      ItemXMLPort.SETDESTINATION(OutStreamQueue);

      ItemXMLPort.EXPORT;

      CREATE(XMLDoc);

      XMLDoc.save(OutStreamQueue);

      OutMsg.Send(0);

      CLEAR(OutMsg);

    END;

    PROCEDURE PopulateResult@1450001(XMLDoc@1450000 : Automation "{F5078F18-C551-11D3-89B9-0000F81FE221} 3.0:{F6D90F11-9C73-11D3-B32E-00C04F990BB4}:'Microsoft XML, v3.0'.DOMDocument";Result@1450001 : Text[250]);

    VAR

      XMLNode@1450003 : Automation "{F5078F18-C551-11D3-89B9-0000F81FE221} 3.0:{2933BF80-7B36-11D2-B20E-00C04F983E60}:'Microsoft XML, v3.0'.IXMLDOMNode";

      XMLDOMMgmt@1450002 : Codeunit 99008516;

    BEGIN

      IF XMLDOMMgmt.FindNode(XMLDoc,'/NavisionObject/Result',XMLNode) THEN

        XMLNode.text := Result;

    END;

    LOCAL PROCEDURE SendCustomer@1450002(CustomerNo@1450000 : Code[20];OutMsg@1450006 : Automation "{F9A57667-8AC5-45C5-9416-99D3955BCAC0} 1.0:{6CB9762C-E61C-4F96-BA34-8B20D3A5B46E}:'Navision Communication Component version 2'.OutMessage");

    VAR

      Customer@1450003 : Record 18;

      CustomerXMLPort@1450002 : XMLport 80000;

      OutStreamQueue@1450004 : OutStream;

      XMLDoc@1450001 : Automation "{F5078F18-C551-11D3-89B9-0000F81FE221} 3.0:{F6D90F11-9C73-11D3-B32E-00C04F990BB4}:'Microsoft XML, v3.0'.DOMDocument";

    BEGIN

      OutStreamQueue := OutMsg.GetStream();

      Customer.SETRANGE("No.",CustomerNo);

      CustomerXMLPort.SETTABLEVIEW(Customer);

      CustomerXMLPort.SETDESTINATION(OutStreamQueue);

      CustomerXMLPort.EXPORT;

      CREATE(XMLDoc);

      XMLDoc.save(OutStreamQueue);

      OutMsg.Send(0);

      CLEAR(OutMsg);

    END;

    LOCAL PROCEDURE FindNode@3(RootNode@1020 : Automation "{F5078F18-C551-11D3-89B9-0000F81FE221} 3.0:{2933BF80-7B36-11D2-B20E-00C04F983E60}:'Microsoft XML, v3.0'.IXMLDOMNode";NodePath@1000 : Text[250];VAR ReturnedNode@2030 : Automation "{F5078F18-C551-11D3-89B9-0000F81FE221} 3.0:{2933BF80-7B36-11D2-B20E-00C04F983E60}:'Microsoft XML, v3.0'.IXMLDOMNode") : Boolean;

    BEGIN

      ReturnedNode := RootNode.selectSingleNode(NodePath);

      IF ISCLEAR(ReturnedNode) THEN

        EXIT(FALSE)

      ELSE

        EXIT(TRUE);

    END;

    EVENT ComCom@1450001::MessageReceived@1(VAR InMessage@1450000 : Automation ":{00020400-0000-0000-C000-000000000046}:''.IDISPATCH");

    VAR

      Item@1450012 : Record 27;

      SalesHeader@1450010 : Record 36;

      InMsg@1450003 : Automation "{F9A57667-8AC5-45C5-9416-99D3955BCAC0} 1.0:{D184D0AC-61C9-4AC1-B537-0D28C277FEDE}:'Navision Communication Component version 2'.InMessage";

      XMLDoc@1450006 : Automation "{F5078F18-C551-11D3-89B9-0000F81FE221} 3.0:{F6D90F11-9C73-11D3-B32E-00C04F990BB4}:'Microsoft XML, v3.0'.DOMDocument";

      XMLNode@1450005 : Automation "{F5078F18-C551-11D3-89B9-0000F81FE221} 3.0:{2933BF80-7B36-11D2-B20E-00C04F983E60}:'Microsoft XML, v3.0'.IXMLDOMNode";

      InStreamQueue@1450002 : InStream;

      XMLChar@1450007 : Text[1];

      XMLText@1450004 : Text[1024];

      UserName@1450014 : Text[1024];

      CompanyName@1450013 : Text[1024];

      FunctionName@1450008 : Text[1024];

      ParameterList@1450009 : Text[1024];

      DocumentType@1450011 : Option;

      Company@1450016 : Record 2000000006;

      Customer@1450017 : Record 18;

    BEGIN

      InMsg := InMessage;

      InStreamQueue := InMsg.GetStream();

      REPEAT

        InStreamQueue.READTEXT(XMLChar);

        XMLText := XMLText + XMLChar;

      UNTIL InStreamQueue.EOS;

      CREATE(XMLDoc);

      XMLDoc.loadXML(XMLText);

      IF FindNode(XMLDoc,'/NavisionObject/UserName',XMLNode) THEN

        UserName := XMLNode.text;

      IF FindNode(XMLDoc,'/NavisionObject/CompanyName',XMLNode) THEN

        CompanyName := XMLNode.text;

      IF FindNode(XMLDoc,'/NavisionObject/FunctionName',XMLNode) THEN

        FunctionName := XMLNode.text;

      IF FindNode(XMLDoc,'/NavisionObject/ParameterList',XMLNode) THEN

        ParameterList := XMLNode.text;

      CASE FunctionName OF

        'GetCustomer':

          BEGIN

            IF Customer.GET(SELECTSTR(1,ParameterList)) THEN

              SendCustomer(Customer."No.",InMsg.CreateReply)

            ELSE BEGIN

              PopulateResult(XMLDoc,Text007);

              ReturnOriginalMessage(InMsg.CreateReply,XMLDoc);

            END;

          END;

        'BlockCustomer':

          BEGIN

            IF Customer.GET(SELECTSTR(1,ParameterList)) THEN BEGIN

              Customer.VALIDATE(Blocked,Customer.Blocked::All);

              IF Customer.MODIFY(TRUE) THEN BEGIN

                PopulateResult(XMLDoc,Text008);

                ReturnOriginalMessage(InMsg.CreateReply,XMLDoc);

              END;

            END ELSE BEGIN

              PopulateResult(XMLDoc,Text007);

              ReturnOriginalMessage(InMsg.CreateReply,XMLDoc);

            END;

          END;

      END;

      InMsg.CommitMessage;

      CLEAR(InMsg);

    END;

    BEGIN

    END.

  }

}

Modify Codeunit 1 ApplicationManagement as followed
PROCEDURE NASHandler(NASID : Text[260])

…

IF CGNASStartedinLoop = FALSE THEN

  CASE Parameter OF

    'NAVISIONSERVICE': // Add this line

      CODEUNIT.RUN(CODEUNIT::RunNavisionProcess); // Add this line

    'MAILLOG':

      CODEUNIT.RUN(CODEUNIT::"E-Mail Dispatcher");

…

Start the Navision Application Server with the following Start-Up Parameter: NAVISIONSERVICE

You should receive the following error in the Application log of the Event Viewer: “The call to member OpenReceiveQueue failed. MSMQQueueInfo returned the following message: The queue does not exist, or you do not have sufficient permissions to perform the operation.”

This message will appear as long as the web service (below) is not created. The web service will create the required queue in Microsoft Message Queuing.

Creating a web service in Visual Studio 2005

In Visual Studio 2005, go to File / New / Web-Site… and select ASP.NET Web Service
Use the following parameters:
· Location: HTTP http://localhost/NavisionWebServices

· Language: Visual Basic

In the App_Code/Service.vb file of the project, add the following declarations

Imports System.Xml

Imports System.Xml.Serialization

Imports System.Messaging (Go to Website / Add Reference… and select System.Messaging from the .NET tab)

In the App_Code/Service.vb file of the project, add the following web methods to the Public Class section

<WebMethod()> Public Function RunNavision(ByVal CompanyName As String, ByVal FunctionName As String, ByVal ParameterList As String) As XmlElement

    Dim QueueToNavision As String = ".\private$\" + CompanyName + " 1"

    Dim QueueFromNavision As String = ".\private$\" + CompanyName + " 2"

    Dim MyNavisionObject1 As NavisionObject = New NavisionObject

    Dim MyNavisionObject2 As NavisionObject = New NavisionObject

    Dim MyNavisionErrorMsg As NavisionErrorMsg = New NavisionErrorMsg

    Dim MyQueue1 As MessageQueue = New MessageQueue(QueueToNavision)

    Dim MyQueue2 As MessageQueue = New MessageQueue(QueueFromNavision)

    Dim MyMessage1 As Message

    Dim MyMessage2 As Message

    Dim MyXmlDocument As XmlDocument

    Dim MyXmlElement1 As XmlElement

    Dim MyXmlElement2 As XmlElement

    Dim MyFormatter1 = New XmlMessageFormatter(New Type() {GetType(NavisionErrorMsg)})

    If Not MyQueue2.Exists(QueueFromNavision) Then

        MyQueue2.Create(QueueFromNavision)

        MyQueue2.Label = QueueFromNavision

        MyQueue2.Authenticate = 0

        MyQueue2.MessageReadPropertyFilter.CorrelationId = True

        MyQueue2.UseJournalQueue = True

    End If

    If Not MyQueue1.Exists(QueueToNavision) Then

        MyQueue1.Create(QueueToNavision)

        MyQueue1.Label = QueueToNavision

        MyQueue1.Authenticate = 0

        MyQueue1.UseJournalQueue = True

    End If

    MyNavisionObject1.UserName = Me.User.Identity.Name()

    MyNavisionObject1.CompanyName = CompanyName

    MyNavisionObject1.FunctionName = FunctionName

    MyNavisionObject1.ParameterList = ParameterList

    MyNavisionObject1.Result = ""

    MyMessage1 = New Message(MyNavisionObject1)

    MyMessage1.Label = "Navision MSMQ-BA"

    MyMessage1.ResponseQueue = MyQueue2

    MyQueue1.Send(MyMessage1)

    MyXmlDocument = New XmlDocument

    MyXmlElement1 = MyXmlDocument.CreateElement("error", "description", "urn:errors")

    Try

        MyMessage2 = MyQueue2.PeekByCorrelationId(MyMessage1.Id, New TimeSpan(0, 1, 0))

    Catch

        MyXmlElement1.InnerText = "Application Server Does Not Respond"

        Return MyXmlElement1

    End Try

    If MyFormatter1.CanRead(MyMessage2) = True Then

        MyMessage2 = MyQueue2.ReceiveById(MyMessage2.Id)

        MyMessage2.Formatter = New XmlMessageFormatter(New Type() {GetType(NavisionErrorMsg)})

        MyNavisionErrorMsg = CType(MyMessage2.Body, NavisionErrorMsg)

        MyXmlElement1.InnerText = MyNavisionErrorMsg.ErrorMessage

        Return MyXmlElement1

    Else

        MyMessage2 = MyQueue2.ReceiveById(MyMessage2.Id)

        MyMessage2.Formatter = New XmlMessageFormatter(New Type() {GetType(XmlElement)})

        MyXmlElement2 = MyMessage2.Body

        Return MyXmlElement2

    End If

End Function

Go to Website / Add New Item… and select Class
Use the following parameters:
· Name: NavisionErrorMsg.vb

· Language: Visual Basic

In the App_Code/ NavisionErrorMsg.vb file of the project, add the following declaration

Imports System.Xml.Serialization
In the App_Code/ NavisionErrorMsg.vb file of the project, replace the Public Class section with the following VB code

<XmlRoot(Namespace:="urn:www-Navision-com:NavisionErrorMsg.xml", ElementName:="NavisionErrorMsg")> _

Public Class NavisionErrorMsg

    Protected MyErrorNo As String

    Protected MyErrorMessage As String

    Protected MyErrorSource As String

    Protected MyBusAdapterId As String

    <XmlElement(ElementName:="ErrorNo")> _

    Public Property ErrorNo() As String

        Get

            Return Me.MyErrorNo

        End Get

        Set(ByVal Value As String)

            Me.MyErrorNo = Value

        End Set

    End Property

    <XmlElement(ElementName:="ErrorMessage")> _

    Public Property ErrorMessage() As String

        Get

            Return Me.MyErrorMessage

        End Get

        Set(ByVal Value As String)

            Me.MyErrorMessage = Value

        End Set

    End Property

    <XmlElement(ElementName:="ErrorSource")> _

    Public Property ErrorSource() As String

        Get

            Return Me.MyErrorSource

        End Get

        Set(ByVal Value As String)

            Me.MyErrorSource = Value

        End Set

    End Property

    <XmlElement(ElementName:="BusAdapterId")> _

    Public Property BusAdapterId() As String

        Get

            Return Me.MyBusAdapterId

        End Get

        Set(ByVal Value As String)

            Me.MyBusAdapterId = Value

        End Set

    End Property

End Class
Go to Website / Add New Item… and select Class
Use the following parameters:
· Name: NavisionNetFunction.vb

· Language: Visual Basic

In the App_Code/ NavisionNetFunction.vb file of the project, add the following functions to the Public Class section
Public Function ConvertCharacters(ByVal MyString As String) As String

    Dim i As Integer

    Dim ReturnString As String

    Dim IdentifiersToConvert As String

    IdentifiersToConvert = ".""\/'"

    ReturnString = ""

    For i = 0 To MyString.Length - 1

        If InStr(1, IdentifiersToConvert, MyString.Chars(i), CompareMethod.Text) <> 0 Then

            ReturnString = ReturnString + "_"

        Else

            ReturnString = ReturnString + MyString.Chars(i)

        End If

    Next

    Return (ReturnString)

End Function

Go to Website / Add New Item… and Select Class
Use the following parameters:
· Name: NavisionObject.vb

· Language: Visual Basic

In the App_Code/ NavisionObject.vb file of the project, add the following declaration

Imports System.Xml.Serialization

In the App_Code/ NavisionObject.vb file of the project, replace the Public Class section with the following VB code

<XmlRoot(Namespace:="urn:www-Navision-com:NavisionObject.xml", ElementName:="NavisionObject")> _

Public Class NavisionObject

    Protected MyUserName As String

    Protected MyCompany As String

    Protected MyFunction As String

    Protected MyParameter As String

    Protected MyResult As String

    <XmlElement(ElementName:="UserName")> _

    Public Property UserName() As String

        Get

            Return Me.MyUserName

        End Get

        Set(ByVal Value As String)

            Me.MyUserName = Value

        End Set

    End Property

    <XmlElement(ElementName:="CompanyName")> _

    Public Property CompanyName() As String

        Get

            Return Me.MyCompany

        End Get

        Set(ByVal Value As String)

            Me.MyCompany = Value

        End Set

    End Property

    <XmlElement(ElementName:="FunctionName")> _

    Public Property FunctionName() As String

        Get

            Return Me.MyFunction

        End Get

        Set(ByVal Value As String)

            Me.MyFunction = Value

        End Set

    End Property

    <XmlElement(ElementName:="ParameterList")> _

    Public Property ParameterList() As String

        Get

            Return Me.MyParameter

        End Get

        Set(ByVal Value As String)

            Me.MyParameter = Value

        End Set

    End Property

    <XmlElement(ElementName:="Result")> _

    Public Property Result() As String

        Get

            Return Me.MyResult

        End Get

        Set(ByVal Value As String)

            Me.MyResult = Value

        End Set

    End Property

End Class
Go to Build / Build Solution

To test the solution, enter http://localhost/NavisionWebServices/Service.asmx in IIS. The operations supported by the web service should be displayed. You can click the RunNavision operation, specify the following parameters and click Invoke to test the functionality:
· CompanyName: CRONUS International Ltd.
· FunctionName: GetCustomer

· ParameterList: 50000
An XML document must be returned.

The first time the web service is called, it creates the appropriate queues in Microsoft Message Queuing to handle the communications with the Navision Application Server. You may have to change manually the permissions assigned to the queues. Go to Control Panel / Administrative Tools / Computer Management. Expand Services and Applications / Message Queuing / Private Queues. Right click on the name of the queues. Go to Properties and review the content of Security tab.

Calling the web service from Microsoft Word (Customer Credit Limit.doc embedded)

[image: image1.emf]Customer Credit  Limit.doc


(Double-click to open)
Go to Tools / Macro / Security… and set “Low”

Go to View / Toolbars and select Navision

Click Connect to Navision Customer in the Navision toolbar

Enter a Navision Customer No. (“50000” for example).

· The Get Customer Information button replaces the bookmarks in the document with the values of the corresponding fields in the Navision Customer Card. To view the list of bookmarks in the document, go to Insert / Bookmark…
· The Block Customer in Navision button will set the Blocked field of the Navision Customer Card to “All”
[image: image6.png]



Navision and Microsoft Word

5 The Ring

Westminster

GB-W2 8HG London

Great Britain


Phone No.: 0666-666-6666

Fax No.: 0666-666-6660

Subject: Credit Limit


Monday, March 13, 2006

Dear ,

This letter is a friendly reminder to inform you that your Credit Limit of   GBP has been reached. The current balance to date is  GBP.


Best regards,

The CRONUS team













 







Phone No.: 



Fax No.: 












